
Overview

➙ Equality
➙ Parametricity
➙ Type classes
➙ generic = type-indexed
➙ Type equalities
➙ Encoding generic functions



Equality

The problem

Given two values of the same type, decide whether they are equal or
not!

Two questions

➙ Which type would that function have?
➙ How can the function be implemented?



The type of equality

The problem description does not involve a specific type.
Therefore,

equal :: a → a → Bool

which in fact means

equal :: ∀a.a → a → Bool

seems reasonable.



How to implement equality

Given a datatype, it is easy . . .

For natural numbers

data Nat = Zero | Succ Nat

equalNat Zero Zero = True
equalNat (Succ a) (Succ b) = equalNat a b
equalNat = False

For trees of boolean values

data TruthTree = Leaf Bool | Node TruthTree TruthTree

equalTruthTree (Leaf a) (Leaf b) = equalBool a b
equalTruthTree (Node s1 s2) (Node t1 t2) = equalTruthTree s1 t1

∧ equalTruthTree s2 t2
equalTruthTree = False



For all datatypes

Is there an algorithm, expressible in Haskell, to implement generic
equality without knowledge of the datatype?

In other words: can we write

equal :: ∀a.a → a → Bool

?

Answer

No.



Parametricity

The parametricity theorem formalizes the following idea:

A parametrically polymorphic type argument cannot be
inspected/modified/deconstructed in any way!

∀a.a → a → Bool

If such a function cannot inspect the two arguments it gets, then it
necessarily must return a constant Bool, either always True or always
False.



Parametricity – continued

The parametricity theorem is a meta-theorem:
For each parametrically polymorphic datatype, we get a theorem
(the free theorem) that holds for all functions of this datatype.

For

f :: ∀a.a → a → Bool

we get that if a :: a → a′, then

∀x y. f x y = f (a x) (a y)

One can immediately see that f 6≡ equal.



The difference to reverse

Isn’t reverse also changing the elements of the list? After all, the
argument list is completely reversed. Still, we can certainly write a
parametrically polymorphic function

reverse :: ∀a.[a ] → [a ]

in Haskell.

Answer

We are not changing the elements. We are just modifying the list
around the elements.



Excursion: Parametricity explored

How do we get from a type to a theorem? And what does
parametricity tell us for reverse?

Read types as relations!

Theorem (Parametricity Theorem)
If f :: t, then (f , f ) ∈ R(t).

Here, R(t) is a relation based on the type.



Interpreting types as relations

➙ Constant types (such as, in our example, Bool), are interpreted
as the identity relation.

➙ The function arrow is lifted to a function on relations: it takes
to relations A and B to a relation A → B.

(f , f ′) ∈ A → B ⇐⇒ (∀x x′ .(x, x′) ∈ A =⇒ (f x, f ′ x′) ∈ B)

➙ The quantifier is also interpreted: if F (A) is a relation
involving the relation A, then

(x, x′) ∈ λ∀ ·A.F (A) ⇐⇒ (forall relations A.(x, x′) ∈ F (A))

For f :: ∀a.a → a → Bool, the parametricity theorem reads:

(f , f ) ∈ ∀ ·A.A → A → Id

or

forall relations A.(f , f ) ∈ A → A → Id



Applying the definitions

The rest are simple transformations: from

forall relations A.(f , f ) ∈ A → A → Id

we get the equivalent

forall relations A.∀x x′ .(x, x′)‘ in ‘A =⇒ (f x, f x′)‘ in ‘A → Id

and then

forall relations A.∀x x′ y y′ .(x, x′)‘ in ‘A ∧ (y, y′)‘ in ‘A =⇒ (f x y, f x′ y′)‘ in ‘Id

If we assume that A is a function, then x′ and y′ can be replaced by
a x and a y:

forall functions a.∀x y.f x y = f (a x) (a y)

Remark

For the type of reverse, we get a free theorem saying that reverse
commutes with map.



Escaping from parametricity: type classes

Haskell has an equality function, but it is not parametrically
polymorphic. It has type

∀a.(Eq a) ⇒ a → a → Bool

Type classes have been introduced to provide ad-hoc polymorphism.
➙ Use one function name for different functions with a related

type.
➙ Here: Use one equality function for different equality functions

on different datatypes.
➙ The functions on the different types are not enforced to be

related.
➙ For a function such as equality, that means that instances have

to be written for each datatype although there is a general
algorithm to provide instances.

➙ Yes, there is deriving in Haskell to generate a few functions
automatically, but only for a very limited and fixed set of
functions/classes!



Dictionary translation

The class constraints can also be seen as hidden arguments that the
compiler fills in.

∀a.Eq a → a → a → Bool

The type Eq a containins the implementation of the equality function
for that type. It is called a dictionary argument.



Instance rules are a step in the right direction

Haskell provides facilities to generate instances of classes in a
systematic way, such as

instance (Eq a) ⇒ Eq [a ]

This can be seen as a function on dictionaries with type
Eq a → Eq [a ].



Generic programming in context

Ad-hoc overloading

Generic programming

Parametric polymorphism



Idealized generic function

An ideal version of equality would get a datatype as argument and
could do something with it.

equal :: (t ::∗) → t → t → Bool
equal t = typecase t of

a pair of two types a and b →
λ(a1 , b1) (a2 , b2) → equal a a1 a2 ∧ equal b b1 b2

a disjunct union of two types a and b →
λ(Left a1) (Left a2) → equal a a1 a2
λ(Right b1) (Right b2) → equal b b1 b2
λ → False

. . .

The type argument is only useful if there is a typecase with sensible
patterns to match against.



Generic programming in Haskell

➙ Simulate type arguments.
➙ Represent Haskell datatypes in a uniform way.



First approach: Universal datatype

We could have one datatype Type to represent all datatypes.

data Type = Pair Type Type
| Union Type Type
| Unit
| Int
| . . .

Equality would then get type

equal :: Type → t → t → Bool

We lose the connection between the type argument and the two
value arguments.

Variation:

equal :: Dynamic → Dynamic → Bool

If we add the value itself to the type representation, we lose the
condition that the two arguments have to be of the same type.



Building a type representation type

The situation improves if we keep the original type around.

data Type t = Pair (Type a) (Type b)
| Union (Type a) (Type b)
| Unit
| Int
| . . .

The type of equality would be close to the “ideal type”:

equal :: Type t → t → t → Bool
equal :: (t :: ∗) → t → t → Bool

However, this is not enough:
➙ There is a connection between the red variables and the type

argument that is not captured.
➙ There is no way to enforce the structure of the two arguments

based on the pattern match on the type argument.



Details

We can better identify the difficulty when we try to actually
implement equality this way:

equal :: Type t → t → t → Bool
equal t = case t of

Pair a1 b2 →
λ(a1 , b1) (a2 , b2) → equal a a1 a2 ∧ equal b b1 b2

. . .

equal :: (t ::∗) → t → t → Bool
equal t = typecase t of

a pair of two types a and b →
λ(a1 , b1) (a2 , b2) → equal a a1 a2 ∧ equal b b1 b2

a disjunct union of two types a and b →
λ(Left a1) (Left a2) → equal a a1 a2
λ(Right b1) (Right b2) → equal b b1 b2

. . .



Encoding type constraints

We need a way to encode equations between types, and to enforce
these equations.

data Type t = ∃ a b. Pair (Type a) (Type b) (t ≡ (a, b))
| ∃ a b. Union (Type a) (Type b) (t ≡ Either a b)
| Unit (t ≡ Unit)
| Int (t ≡ Int)
| . . .

Think of ≡ as if it was just another parametrized datatype. It could
alternatively be written as

data Equal a b = . . .

Assuming there are conversion functions between equal types, we
can implement the cases of the generic function successfully:

from ::(a ≡ b) → (a → b)
to ::(a ≡ b) → (b → a)



Generic equality
data Type t = ∃ a b.Pair (Type a) (Type b) (t ≡ (a, b))

| ∃ a b.Union (Type a) (Type b) (t ≡ Either a b)
| Unit (t ≡ ())
| Int (t ≡ Int)
| . . .

equal :: Type t → t → t → Bool
equal t = case t of

Pair a b conv →
λpair1 pair2 →

case (from conv pair1 , from conv pair2) of
((a1 , a2), (b1 , b2)) → equal a a1 a2 ∧ equal b b1 b2

Union a b conv →
λunion1 union2 →

case (from conv union1 , from conv union2) of
( Left a1 , Left a2 ) → equal a a1 a2
( Right b1 , Right b2 ) → equal b b1 b2

→ False
. . .



Implementing type constraints

An intriguing possiblity is to use the following type

data a ≡ b = Proof{ apply :: ∀f .f a → f b }

This type guarantees equality of a and b. It captures the notion of
“extensional” equality mentioned earlier: If every
property/observation of a is also one of b, then a and b must be
equal.

newtype Arr a b = Arr { unArr :: a → b }
newtype Rev a b = Rev { unRev :: b → a }

from conv = unArr (apply conv (Arr id))
to conv = unRev (apply conv (Rev id))



Embedding-projection pairs

A slightly less restrictive type is also an option:

data a ≡ b = EP{ from :: a → b, to :: b → a}

This type does not guarantee that the types are equal or isomorphic.

However, embedding projection pairs are of great help with our
remaining problem: creating suitable type representations.



Type representations for real datatypes

Here, manual work is needed for each datatype, but only once!

refl = EP{ from = id, to = id}

repUnit = Unit refl
repUnion a b = Union a b refl
repPair a b = Pair a b refl

repNat = Union repUnit repNat epNat

epNat :: Nat ≡ (Either () Nat)
epNat = EP{ from· = fromNat, to· = toNat}

fromNat Zero = Left ()
fromNat (Succ n) = Right n

toNat (Left ()) = Zero
toNat (Right n) = Succ n



Generating embedding-projection pairs

For each datatype a, the following is needed:

repa :: Type a — making use of epa
epa :: a ≡ r — for some suitable r

The type r makes use of (), Either, and (,) to break down the multiple
alternatives and fields into applications of simple constructors.

data TruthTree = Leaf Bool | Node TruthTree TruthTree

epTruthTree:: TruthTree ≡ (Either Bool (TruthTree, TruthTree))
epTruthTree = EP{ from = fromTruthTree, to = toTruthTree}

fromTruthTree (Leaf b) = Left b
fromTruthTree (Node t1 t2) = Right (t1 , t2)

toTruthTree (Left b) = Leaf b
toTruthTree (Right (t1 , t2)) = Node t1 t2



Type constraints are powerful!

To give an expression what more is possible using type constraints
in datatypes, consider map! Informally,

➙ the function map is only defined on type constructors (on
constant types, it can be seen as the identity);

➙ if we view a parametrized datatype as a container for elements
of the parameter type, then we ask for a suitable function
converting values of this type into something else;

➙ we then traverse the “structure” of the container and apply the
function to all elements of this type;

➙ if a type constructor has multiple parameters, we need
multiple mapping functions.

mapInt :: Int → Int
map[ ] :: ∀a b. (a → b) → [a ] → [b ]
mapEither :: ∀a b c d. (a → c) → (b → d) → Either a b → Either c d

We can write such a map, with a variable number of arguments,
using a datatype with type constraints.



Using a type as a relation

map :: ∀r.Map r → r

The type Map establishes a relation between type representations
and result types:

data Map r = Unit (r ≡ () → ())
| ∀a b c d. Pair (r ≡ (a → c) → (b → d) → (a, b) → (c, d))
| ∀a b c d. Union (r ≡ (a → c) → (b → d) → Either a b → Either c d)
| . . .

We additionally provide the lambda calculus operations on the
datatype, i.e. we add cases for abstraction, application, and
variables:

| ∀a b. Lam (Map a → Map b) (r ≡ a → b)
| ∀a b. App (Map (a → b)) (Map a) (r ≡ b)
| ∀a b. Var t



Representing types

Besides the constant type representations, we now also get
application and abstraction on type representations.

repUnit = Unit refl
repPair = Pair refl
repUnion = Union refl

a $$ b = App a b refl
lambda t = Lam t refl

Together with generated representations for “real” datatypes, we
can now build complex type expressions and get a map of the
appropriate type:

map (lambda (λx → rep[ ] $$ (rep[ ] $$ x))) :: ∀a b.(a → b) → [ [a ] ] → [ [b ] ]



The definition

We can make use of the standard definitions for map on the three
base type constructors:

map (Unit conv) = to conv map()
map (Pair conv) = to conv map(,)
map (Union conv) = to conv mapEither

The remaining cases are independent of the map function and
reappear in other, similar generic functions. They can be abstracted
out.

map (Lam t conv) = to conv (λx → map (t (Var x)))
map (App t1 t2 conv) = to conv ((map t1) (map t2))
map (Var x) = x

Do you recognize Ralf Hinze’s “MPC”-style generics here?



Comparison with type classes

Multi-parameter type classes with functional dependencies can be
used to achieve many of the things that have been done with type
constraints in datatypes here. Some differences:

➙ Type classes are extensible, datatypes are closed. Sometimes
extensibility may be wanted, for instance, to assign a special
behaviour to a specific datatype.

➙ Type classes can be used to apply some of the coercion
functions automatically and to make the type argument
implicit.

➙ The function definitions look much more natural using
datatypes, because we can perform pattern matching on the
type argument. With classes, function definitions are scattered
over several instance definitions.

➙ Tricks like implementing the map function require heavy use of
functional dependencies, undecidable instances, and so on.
Here, we use existential datatypes, otherwise its plain Haskell.



Conclusions

➙ Generic functions are more expressive than parametrically
polymorphic functions.

➙ Generic functions allow to capture the ideas of algorithms that
are based on datatype structure, and thus increases reusability
of code.

➙ Generic programming, although not supported by Haskell, can
be approximated and simulated.

➙ In the approach that has been discussed, two main
disadvantages remain: the generation of embedding-projection
pairs for datatypes has to be done by hand, and the application
of coercion functions is quite annoying.

➙ These disadvantages are removed by a language extension
such as Generic Haskell.


